# Calculations used in Chapter 5: Core Ionization

James Avery (avery@diku.dk)

April 7, 2008

Core ionization energies are the minimum energies needed to remove an electron from a 1s-orbital of an atom. These energies are cumbersome to calculate by conventional methods, but easy using generalized Sturmians.

We will start by discussing the general features of core ionization energies using the large-$Z$ approximation discussed in Chapter 3. This approximation is equivalent to using a basis set limited to configurations that are degenerate if interelectron repulsion is neglected. These configurations, which we call an $\mathcal{R}$-block, all correspond to a particular value of

$$\mathcal{R}_\nu = \sqrt{\frac{1}{n_1^2} + \frac{1}{n_2^2} + \cdots + \frac{1}{n_N^2}} \tag{1}$$

In the large-$Z$ approximation the energy of an atomic state is given by

$$E_\kappa = -\frac{1}{2}\left(\mathcal{R}_\nu Z - |\lambda_\kappa|\right)^2 \tag{2}$$

where $\lambda_\kappa$ is the corresponding eigenvalue for the $\mathcal{R}$-block of the interelectron repulsion matrix.

## 1 Initialization

In[1] :=

We begin by loading the libraries that we will use:

```
(* For rapid calculation of repulsion-matrix *)
Get["Sturmian`binary`"];
Get["Sturmian`M`misc`"];
(* For constructing basis sets that obey the Pauli principle *)
Get["Sturmian`M`pauli`"];
(*For automatic identification of the symmetry of eigenstates *)
Get["Sturmian`M`identify`"];
Get["Sturmian`M`format`"];
Off[General::spell];
```

## 2 Set-up and auxiliary functions

In order to calculate the $\mathcal{R}$-blocks appropriate for a core ionization calculation, we set `o1`, `o2` and `o3` to the sets of one-electron quantum numbers with $n = 1$, $n = 2$ and $n = 3$ respectively, appropriate for lowest shells when $1 \le N \le 18$. We then generate `bigbases`, a list of all possible $N$-electron configurations ($2 \le N \le 10$) allowed by the Pauli principle and restricted to the $n = 1$ and $n = 2$ shells.

In[2] :=

```
{o1, o2, o3} = oneelectronstates[#, #] & /@ {1, 2, 3};
o12 = Join[o1, o2];

(* N is a reserved word in Mathematica (numerical evaluation), so in
  the code that follows, Nel is used in its place. *)
bigbases = Table[binomialchoices[o12, Nel], {Nel, 2, 10}];
```

Next, we compute bases for the ground state $\mathcal{R}$-blocks for $2 \leq N \leq N_{\max}$ and for the ionized system for $1 \leq N' \leq N_{\max} - 1$:

For the ground state $\mathcal{R}$-blocks (`rblocks0`), we have two cases: If $2 \leq N \leq 10$, we select from `bigbases` those $N$-electron configurations for which the first two electrons are in the $n = 1$ shell and the remainder in the $n = 2$ shell. If $10 < N \leq 18$, the $n = 1$ and $n = 2$ shells are fully occupied, and the remaining electrons are placed in the $n = 3$ shell. The $\mathcal{R}$-blocks appropriate for describing the ionized states (`rblocksi`) are constructed from the ground state blocks by removing an electron from the 1s-shell.

In[3] :=

```
Nmax = 11;
N10core = {{1, 0, 0, 1}, {1, 0, 0, -1}, {2, 0, 0, 1}, {2, 0,
   0, -1}, {2, 1, -1, 1}, {2, 1, 0, 1}, {2, 1, 1, 1}, {2,
   1, -1, -1}, {2, 1, 0, -1}, {2, 1, 1, -1}};

rblock0[n_] := Module[{threeblock},
  If[n <= 10, Return[Select[bigbases[[n - 1]], (#[[1, 1]] == 1) && (#[[2, 1]] == 1) &]]];
  threeblock = Join[N10core, #] & /@ binomialchoices[o3, n - 10];
  Return[threeblock];
];

rblocki[n_] := Drop[#, 1] & /@ rblock0[n];

rblocksi = Table[rblocki[n], {n, 2, Nmax}];
rblocks0 = Table[rblock0[n], {n, 2, Nmax}];
```

The dimensions of the $\mathcal{R}$-blocks should be `Binomial[8,N-2]` for $2 < N \leq 10$ and `Binomial[18,N-10]` for $10 < N \leq 18$, corresponding to the number of electrons in the open subshell. We can check by inspection that this is the case.

In[4] :=

```
Length /@ rblocks0
Table[Binomial[8, Nel - 2], {Nel, 2, 10}]
Table[Binomial[18, Nel - 10], {Nel, 11, Nmax}]
```

Out[4] :=

```
{1,8,28,56,70,56,28,8,1,18}
{1,8,28,56,70,56,28,8,1}
{18}
```

# 3 Finding the Interelectron Repulsion Roots

We next compute the two lists of interelectron repulsion-matrices: `Tsi` for the ionized systems and `Ts0` for the ground state systems.

In[5] :=

```
Tsi = repulsionMatrix /@ rblocksi;
Ts0 = repulsionMatrix /@ rblocks0;
```

`classify[Trep,basis,tolerance]` gives us the roots of matrix `Trep` and the symmetries of the corresponding eigenvectors. We are interested in the roots of the $\mathcal{R}$-blocks:

In[6] :=

```
roots = classify[Tsi[[2]], rblocksi[[2]], 10^(-8)]
```

Out[6] :=

```
{{-0.168089,{0,1}},{-0.187719,{0,0}},{-0.201897,{1,1}},{-0.217165,{1,0}}}
```

The symmetries are given as pairs (L,S). To increase readability, we can apply `term2label[]` to show them in the style of standard spectroscopic notation.

In[7] :=

```
{#[[1]], term2label[#[[2]]]} & /@ roots
```

Out[7] :=

$$\{\{-0.168089,\ {}^3S\},\{-0.187719,{}^1S\},\{-0.201897,{}^3P\},\{-0.217165,{}^1P\}\}$$

Tables 5.1 and 5.2 show the roots of the $\mathcal{R}$-blocks of the interelectron repulsion matrices `Tsi` corresponding to the core-ionized system. The roots need only be calculated once and may be used for rough approximations to all the series that they represent.

In[8] :=

```
table51data =
Table[classify[Tsi[[i]], rblocksi[[i]], 10^(-8)], {i, Length[rblocksi]}];

table51data2 = {NumberForm[Abs[#[[1]]], 6], term2label[#[[2]]]} & /@ # & /@ table51data;

(table51data3 = Flatten /@ Transpose[padtable[table51data2, {"", ""}]]) // MatrixForm
```

Out[8] :=

| | | | | | | | | | | | | | | | |
|---|---|---|---|---|---|---|---|---|---|---|---|---|---|---|---|
| 0 | $.^2S$ | 0.168089 | $.^3S$ | 0.433936 | $.^2S$ | 0.800757 | $.^3P$ | 1.23703 | $.^4P$ | 1.73489 | $.^5S$ | 2.33058 | $.^4P$ | 2.97391 | $.^3P$ | 3. |
| | | 0.187719 | $.^1S$ | 0.44619 | $.^4P$ | 0.808142 | $.^5P$ | 1.24149 | $.^6S$ | 1.74627 | $.^3S$ | 2.34138 | $.^2P$ | 2.98421 | $.^1P$ | |
| | | 0.201897 | $.^3P$ | 0.462037 | $.^2P$ | 0.813661 | $.^1P$ | 1.2491 | $.^2P$ | 1.75765 | $.^1S$ | 2.35218 | $.^2P$ | 3.07386 | $.^3S$ | |
| | | 0.217165 | $.^1P$ | 0.478048 | $.^2P$ | 0.822231 | $.^3P$ | 1.2544 | $.^4S$ | 1.76736 | $.^3D$ | 2.35472 | $.^2D$ | 3.08709 | $.^1S$ | |
| | | | | 0.502568 | $.^4P$ | 0.836362 | $.^1P$ | 1.26117 | $.^2P$ | 1.76903 | $.^3S$ | 2.3689 | $.^2S$ | | | |
| | | | | 0.510044 | $.^2P$ | 0.850542 | $.^3P$ | 1.26402 | $.^2D$ | 1.77673 | $.^3P$ | 2.41564 | $.^4P$ | | | |
| | | | | 0.516506 | $.^2P$ | 0.859138 | $.^3D$ | 1.26732 | $.^2S$ | 1.77874 | $.^1D$ | 2.42791 | $.^2P$ | | | |
| | | | | 0.530445 | $.^2P$ | 0.875729 | $.^1D$ | 1.27824 | $.^2S$ | 1.78811 | $.^1P$ | 2.44032 | $.^2P$ | | | |
| | | | | 0.533729 | $.^2D$ | 0.883056 | $.^3S$ | 1.28026 | $.^2S$ | 1.80371 | $.^5P$ | 2.4651 | $.^2P$ | | | |
| | | | | 0.573228 | $.^2S$ | 0.883961 | $.^5S$ | 1.29323 | $.^4S$ | 1.81614 | $.^3P$ | 2.55129 | $.^2S$ | | | |
| | | | | | | 0.889988 | $.^3P$ | 1.29665 | $.^4D$ | 1.8286 | $.^1P$ | | | | | |
| | | | | | | 0.896865 | $.^3S$ | 1.31038 | $.^2D$ | 1.8411 | $.^3P$ | | | | | |
| | | | | | | 0.899647 | $.^1S$ | 1.31157 | $.^4P$ | 1.84869 | $.^3D$ | | | | | |
| | | | | | | 0.901665 | $.^1P$ | 1.32424 | $.^2D$ | 1.86332 | $.^1D$ | | | | | |
| | | | | | | 0.90977 | $.^1S$ | 1.32529 | $.^2P$ | 1.86978 | $.^3S$ | | | | | |
| | | | | | | 0.920783 | $.^3D$ | 1.33904 | $.^4S$ | 1.87589 | $.^3P$ | | | | | |
| | | | | | | 0.922674 | $.^3S$ | 1.33916 | $.^2P$ | 1.88441 | $.^1S$ | | | | | |
| | | | | | | 0.933688 | $.^1D$ | 1.35026 | $.^2S$ | 1.88619 | $.^1P$ | | | | | |
| | | | | | | 0.943943 | $.^3P$ | 1.35195 | $.^2D$ | 1.94394 | $.^3P$ | | | | | |
| | | | | | | 0.956847 | $.^1P$ | 1.36146 | $.^2S$ | 1.95532 | $.^1P$ | | | | | |
| | | | | | | | | 1.36687 | $.^2P$ | | | | | | | |
| | | | | | | | | 1.38679 | $.^4P$ | | | | | | | |
| | | | | | | | | 1.39887 | $.^2P$ | | | | | | | |
| | | | | | | | | 1.41094 | $.^2P$ | | | | | | | |
| | | | | | | | | 1.41378 | $.^2D$ | | | | | | | |
| | | | | | | | | 1.44431 | $.^2S$ | | | | | | | |

We'll use the smallest roots from Table 5.1 and 5.2 - corresponding to the most tightly bound states in the large-$Z$ approximations - for our example:

In[9] :=

```
47 smallestroots = First /@ table51data
48 {#[[1]], term2label[#[[2]]]} & /@ smallestroots
```

Out[9] :=

```
{{0, {0, 1/2}}, {-0.16808851188887308, {0, 1}}, {-0.43393577911809267, {0, 1/2}},
 {-0.8007570901960334, {1, 1}}, {-1.2370299672181175, {1, 3/2}}, {-1.7348882125438199, {0, 2}},
 {-2.33058245905515, {1, 3/2}}, {-2.973914348693145, {1, 1}}, {-3.6618129807903506, {0, 1/2}},
 {-4.029874239676303, {0, -1}}}
```

Out[9] :=

```
{{0,²S}, {-0.16808851188887308,³S}, {-0.43393577911809267,²S}, {-0.8007570901960334,³P},
{-1.2370299672181175,⁴P}, {-1.7348882125438199,⁵S}, {-2.33058245905515,⁴P}, {-2.973914348693145,³P},
{-3.6618129807903506,²S}, {-4.029874239676303,³S}}
```

# 4 Computing Core-Ionization Energies

Now we're ready to calculate approximate energies. `Eapprox0[N,Z]` defines the approximate energies of the $N$-electron ground states with nuclear charge $Z$, as defined in Eq. 5.1; `Eapprox1[N-1,Z]` similarly gives approximate energy for the core-ionized states.

In[10] :=

```
53 λ0 = Last /@ Eigenvalues /@ Ts0;
54 r0 = Table[Sqrt[2 + i/4], {i, 0, 8}];
55 λ1 = First /@ smallestroots;
```

```
r1 = Table[Sqrt[1 + i/4], {i, 0, 8}];

Eapprox0[Nel_, Z_] := -1/2 *(Z*r0[[Nel - 1]] + λ0[[Nel - 1]])^2;
Eapprox1[Np_, Z_] := -1/2*(Z*r1[[Np]] + λ1[[Np]])^2;
```

The following are the parameters that enter Equation 5.1; the top number in the brackets is $Z\mathcal{R}_\nu$, while the lower number is $\lambda_\kappa$:

In[11] :=

```
{Table[{Z r0[[i]], λ0[[i]]}, {i, 9}]} // MatrixForm
```

Out[11] :=

$$\left(\begin{pmatrix}\sqrt{2}Z\\-0.441942\end{pmatrix}\ \begin{pmatrix}\frac{3Z}{2}\\-0.68187\end{pmatrix}\ \begin{pmatrix}\sqrt{\frac{5}{2}}Z\\-0.986172\end{pmatrix}\ \begin{pmatrix}\frac{\sqrt{11}Z}{2}\\-1.40355\end{pmatrix}\ \begin{pmatrix}\sqrt{3}Z\\-1.88151\end{pmatrix}\ \begin{pmatrix}\frac{\sqrt{13}Z}{2}\\-2.41491\end{pmatrix}\ \begin{pmatrix}\sqrt{\frac{7}{2}}Z\\-3.02641\end{pmatrix}\ \begin{pmatrix}\frac{\sqrt{15}Z}{2}\\-3.68415\end{pmatrix}\ \begin{pmatrix}2Z\\-4.38541\end{pmatrix}\right)$$

In this way we find that $E_0 \approx -\frac{1}{2}\left(Z\sqrt{2} - 0.441942\right)^2$ when $N = 2$, $E_0 \approx -\frac{1}{2}(3Z/2 - 0.68187)^2$ when $N = 3$, et cetera.

Similarly, the parameters used in Equation 5.3 are:

In[12] :=

```
{Table[{Z r1[[i]], λ1[[i]]}, {i, 9}]} // MatrixForm
```

Out[12] :=

$$\left(\begin{pmatrix}Z\\0\end{pmatrix}\ \begin{pmatrix}\frac{\sqrt{5}Z}{2}\\-0.168089\end{pmatrix}\ \begin{pmatrix}\sqrt{\frac{3}{2}}Z\\-0.433936\end{pmatrix}\ \begin{pmatrix}\frac{\sqrt{7}Z}{2}\\-0.800757\end{pmatrix}\ \begin{pmatrix}\sqrt{2}Z\\-1.23703\end{pmatrix}\ \begin{pmatrix}\frac{3Z}{2}\\-1.73489\end{pmatrix}\ \begin{pmatrix}\sqrt{\frac{5}{2}}Z\\-2.33058\end{pmatrix}\ \begin{pmatrix}\frac{\sqrt{11}Z}{2}\\-2.97391\end{pmatrix}\ \begin{pmatrix}\sqrt{3}Z\\-3.66181\end{pmatrix}\right)$$

Thus we have $E_i \approx -\frac{1}{2}Z^2$ when $N' = N - 1 = 1$, $E_i \approx -\frac{1}{2}\left(Z\sqrt{5}/2 - 0.168089\right)^2$ when $N' = 2$, and so on.

Taking the difference between these energies, we obtain Equation 5.4, which defines the approximate energies $\Delta E[N, Z]$ required to ionize an electron from the core for various values of $N$.

In[13] :=

```
(deltas = Table[Expand[Eapprox1[Nel - 1, Z] - Eapprox0[Nel, Z]], {Nel, 2, 10}]) // MatrixForm
```

Out[13] :=

$$\begin{pmatrix}0.0976563 - 0.625Z + \frac{Z^2}{2}\\0.218347 - 0.834877Z + \frac{Z^2}{2}\\0.392117 - 1.02781Z + \frac{Z^2}{2}\\0.664372 - 1.26822Z + \frac{Z^2}{2}\\1.00491 - 1.50944Z + \frac{Z^2}{2}\\1.41097 - 1.7512Z + \frac{Z^2}{2}\\1.86378 - 1.97693Z + \frac{Z^2}{2}\\2.36441 - 2.20266Z + \frac{Z^2}{2}\\2.91149 - 2.42838Z + \frac{Z^2}{2}\end{pmatrix}$$

Table 5.3 shows $\Delta$E for $2 \le N, Z \le 10$.

In[14] :=

```
table53data = Table[If[Nel < Z, deltas[[Nel]], 0], {Nel, Length[deltas]}, {Z, 2, 10}];
MatrixForm[Map[chew[#, 4] &, table53data, {2}]]
```

Out[14] :=

| | | | | | | | | |
|---|---|---|---|---|---|---|---|---|
| 0.84766 | 2.7227 | 5.5977 | 9.4727 | 14.348 | 20.223 | 27.098 | 34.973 | 43.848 |
| 0 | 2.2137 | 4.8788 | 8.544 | 13.209 | 18.874 | 25.539 | 33.204 | 41.87 |
| 0 | 0 | 4.2809 | 7.753 | 12.225 | 17.697 | 24.17 | 31.642 | 40.114 |
| 0 | 0 | 0 | 6.8232 | 11.055 | 16.287 | 22.519 | 29.75 | 37.982 |
| 0 | 0 | 0 | 0 | 9.9483 | 14.939 | 20.929 | 27.92 | 35.911 |
| 0 | 0 | 0 | 0 | 0 | 13.653 | 19.401 | 26.15 | 33.899 |
| 0 | 0 | 0 | 0 | 0 | 0 | 18.048 | 24.571 | 32.095 |
| 0 | 0 | 0 | 0 | 0 | 0 | 0 | 23.041 | 30.338 |
| 0 | 0 | 0 | 0 | 0 | 0 | 0 | 0 | 28.628 |

The following reproduces Figure 5.1: The energy $\Delta E$ required to remove an electron from the inner shell of an $N$-electron atom or ion is an approximate quadratic function of the nuclear charge $Z$ (Equation 5.4).

In[15] :=

```
fig51lines = Table[Plot[(Eapprox1[Nel - 1, Z] - Eapprox0[Nel, Z])/Z^2,
                 {Z, Nel, 20}], {Nel, 2, 7}];

labels = {Graphics[Text["N=2", {2, .2}]], Graphics[Text["N=7", {7, .267}]]};

figure51 = Show[Join[fig51lines, labels],
              PlotRange -> {0, .5}, AxesOrigin -> {0, 0},
              AxesLabel -> {"Z", "ΔE/Z^2"},
              TextStyle -> {FontSize -> 15, FontFamily -> Times}]
```

Out[15] :=




Finally, for Figure 5.2, we calculate the core ionization energy $\Delta E$ and plot it as a function of electron count $N$ for several values of nuclear charge $Z$. The approximate linear dependence of $N$ is discussed in Chapter 5; it is in fact approximately piecewise linear, with breaks in slope occurring when subshells are filled, i.e. at $N = 2$, $N = 10$, $N = 18$, etc.

In[16] :=

```
lines = Table[
  ListPlot[Table[{Nel, Eapprox1[Nel - 1, Z] - Eapprox0[Nel, Z]}, {Nel, 2, Min[10, Z]}],
           Joined -> True], {Z, 3, 15}
];

dots = Table[
  ListPlot[Table[{Nel, Eapprox1[Nel - 1, Z] - Eapprox0[Nel, Z]}, {Nel, 2, Min[10, Z]}],
           PlotStyle -> PointSize[.013]], {Z, 3, 15}
];

linedots = Table[Show[lines[[i]], dots[[i]]], {i, Length[lines]}];

linelabels = {Graphics[Text["Z=15", {10.55, 78}]],
  Graphics[Text["Z=10", {10.55, 28}]]};

figure52 = Show[Join[linedots, linelabels],
              AxesLabel -> {"N", "ΔE"},
              TextStyle -> {FontSize -> 15, FontFamily -> Times},
              AxesOrigin -> {2, 0}, PlotRange -> {{2, 11.5}, {0, 105}}]
```

Out[16] :=




In Figure 5.2, we are struck by the approximate linear dependence of $\Delta E$ on the number of electrons $N$, a parameter that has very complex effects on the wavefunctions. How can this interesting result be explained? In the next notebook, we will generate the electron density distributions for an isonuclear series in order to explore this question.